
COMPUTER METHODS AND MODELING IN GEOLOGY 
LAKE LEVEL CHANGES IN THE ARID WEST 

 
The parts of this exercise for students are in normal text, whereas answers and 
explanations for faculty are italicized. 
 
Closed-basin lakes in the western United States have been termed “nature’s rain 
gauges” because they respond to changes in precipitation by changing their 
levels.  Many basins in the arid West today contain small lakes, but also contain 
shoreline deposits that indicate these lakes were once much larger.  Basin center 
deposits combined with shoreline materials indicate that lakes oscillated in size 
during the late Pleistocene.  The causes of these oscillations are not yet entirely 
understood, though many believe that lakes expanded because the jet stream 
was forced southward by the large Laurentide ice sheet.  The jet stream (today 
found at the latitude of Washington and Oregon) is the locus of storm activity, 
and computer models of late Pleistocene climate suggest that it split around the 
ice sheet, with its southern track ending up in Nevada and eastern California. 
 
In today’s lab we’ll explore the impact of changes in climate on the level of lakes 
in the Owens River system.  These lakes, which were separated by bedrock sills 
and which were fed by runoff from the eastern flank of the Sierra Nevada 
mountains in California, were headed by Owens Lake.  When Owens Lake filled 
to its maximum level, it overflowed into the China Lake Basin, which in turn 
overflowed into Searles Lake.  During particularly wet periods in the geologic 
past Searles overflowed into Panamint Lake, which ultimately overflowed into 
Manly Lake in Death Valley.  In our modeling effort today we’ll see what 
combinations of runoff and evaporation might have led to Pleistocene lake level 
oscillations. 
 
Readings 
Menking, K.M., and Anderson, R.S., unpublished, A Model of Runoff, 

Evaporation, and Overspill in the Owens River System of Lakes, Eastern 
California. 

 
Exercises 
1) Create a model of the entire chain of lakes. Note that the evaporative outflow 

from each lake depends on the surface area of the lake.  This requires that we 
incorporate into our model a relationship between lake volume and area for 
each lake in the chain. 

STELLA code - provided in the event that you are using an older version of 
STELLA than that we're using or if you have problems downloading and opening 
the model 
 
China_volume(t) = China_volume(t - dt) + (overflow_to_China - evap_from_China - overflow_to_Searles) 
* dt 
INIT China_volume = 0 
 
INFLOWS: 
overflow_to_China = if(Owens_volume>25.e9)then(Owens_volume-25.e9)else(0.0) 
OUTFLOWS: 
evap_from_China = 1.41*China_area 
overflow_to_Searles = if(China_volume>1.2e9)then(China_volume-1.2e9)else(0.0) 
Manly_volume(t) = Manly_volume(t - dt) + (overflow_to_Manly - evap_from_Manly) * dt 
INIT Manly_volume = 0 
 
INFLOWS: 
overflow_to_Manly = if(Panamint_volume>117.e9)then(Panamint_volume -117.e9)else(0.0) 
OUTFLOWS: 
evap_from_Manly = 1.97*Manly_area 
Owens_volume(t) = Owens_volume(t - dt) + (runoff_from_Sierras - evap_from_Owens - 
overflow_to_China) * dt 
INIT Owens_volume = 0 
 
INFLOWS: 
runoff_from_Sierras = 1*3.98e8 
OUTFLOWS: 
evap_from_Owens = 1.34*Owens_area 
overflow_to_China = if(Owens_volume>25.e9)then(Owens_volume-25.e9)else(0.0) 
Panamint_volume(t) = Panamint_volume(t - dt) + (overflow_to_Panamint - evap_from_Panamint - 
overflow_to_Manly) * dt 
INIT Panamint_volume = 0 



 
INFLOWS: 
overflow_to_Panamint = if(Searles_volume>76.0e9)then(Searles_volume -76.0e9)else(0.0) 
OUTFLOWS: 
evap_from_Panamint = 1.65*Panamint_area 
overflow_to_Manly = if(Panamint_volume>117.e9)then(Panamint_volume -117.e9)else(0.0) 
Searles_volume(t) = Searles_volume(t - dt) + (overflow_to_Searles - evap_from_Searles - 
overflow_to_Panamint) * dt 
INIT Searles_volume = 0. 
 
INFLOWS: 
overflow_to_Searles = if(China_volume>1.2e9)then(China_volume-1.2e9)else(0.0) 
OUTFLOWS: 
evap_from_Searles = 1.65*Searles_area 
overflow_to_Panamint = if(Searles_volume>76.0e9)then(Searles_volume -76.0e9)else(0.0) 
China_area = if(China_volume<=0.)then(0.)else 
if(China_volume>=1.2e9)then(C_max_a)else(1.238e8+(0.047279*China_volume)-(1.4788e-
12*(China_volume^2))) 
China_depth = if(China_volume<=0.)then(0.)else 
if(China_volume>=1.2e9)then(C_max_d)else(0.98637+(5.9886e-9*China_volume)-(3.7316e-
19*(China_volume^2))+(1.2192e-29*(China_volume^3))) 
C_max_a = 1.886548e8 
C_max_d = 7.7 
Manly_area = 
if(Manly_volume<0.)or(Manly_volume=0.)then(0.)else(2.1382e8+(0.023849*Manly_volume)-(1.9685e-
13*(Manly_volume^2))+(6.5806e-25*(Manly_volume^3))) 
Manly_depth = if(Manly_volume=0.)then(0.)else(14.782+(2.1584e-9*Manly_volume)-(1.3675e-
20*(Manly_volume^2))+(4.2019e-32*(Manly_volume^3))) 
Owens_area = if(Owens_volume<=0.)then(0.)else 
if(Owens_volume>=25.e9)then(O_max_a)else(75991.*(Owens_volume^0.37636)) 
Owens_depth = if(Owens_volume<=0.)then(0.)else 
if(Owens_volume>=25.e9)then(O_max_d)else(3.5837+(4.009e-9*Owens_volume)-(9.5021e-
20*(Owens_volume^2))+(1.2319e-30*(Owens_volume^3))) 
O_max_a = 5.665292e8 
O_max_d = 63.7 
Panamint_area = if(Panamint_volume<=0.)then(0.)else 
if(Panamint_volume>=117.e9)then(P_max_a)else(1.1632e8+(0.0085949*Panamint_volume)-(2.4725e-
14*(Panamint_volume^2))) 
Panamint_depth = if(Panamint_volume<=0.)then(0.)else 
if(Panamint_volume>=117.0e9)then(P_max_d)else(11.93+(5.4291e-9*Panamint_volume)-(4.6059e-
20*(Panamint_volume^2))+(1.7191e-31*(Panamint_volume^3))) 
P_max_a = 7.839459E+08 
P_max_d = 292.0 
Searles_area = if(Searles_volume<=0.)then(0.)else 
if(Searles_volume>=7.6e10)then(S_max_a)else(1.691e8+(0.008888*Searles_volume)-(3.4502e-
14*(Searles_volume^2))) 



Searles_depth = if(Searles_volume<=0.)then(0.)else 
if(Searles_volume>=76.0e9)then(S_max_d)else(6.3665+(4.5202e-9*Searles_volume)-(4.2614e-
20*(Searles_volume^2))+(2.1146e-31*(Searles_volume^3))) 
S_max_a = 6.460949E+08 
S_max_d = 196.6 
 
In the Classes>Geo365 folder on your computer you’ll find a file called 
“hypsometry_data.txt” that contains area/volume/depth relationships for each 
lake in the chain.  Create a scatter graph of area as a function of volume for 
Owens Lake and then apply a curve fit to this graph to determine the function 
that relates these two variables.  Use this relationship in your model.  Note that 
your relationship will not be linear, but instead a polynomial or power of some 
sort.  Try to get the best possible curve fit between the two variables. 
 
2) Repeat step 1 for the other 5 lakes in the chain. 
 
The data are available here on the Vassar blackboard site in the file called 
"hypsometry_data.txt."  Each student will likely use a different hypsometry equation, so 
you need to have some tolerance for variability in model outputs from student to student.   
 
Many students will go crazy and try to use polynomials of degree 5 or higher.  In 
general, most of the area/volume/depth relationships work fine with polynomials of order 
2-3. 
 
Since each lake, except Manly, has a spillway, the maximum surface area that each lake 
can attain is limited.  This limitation must be expressed in the area converter that is fed 
by the lake volume reservoir.  For example, for Owens lake, the following expression is 
used: 
 
Owens_area= 
if(Owens_volume<=0.)then(0.)else 
if(Owens_volume>=25.e9)then(O_max_a)else(75991.*(Owens_volume^0.37636)) 
 
This expression sets the area to be zero when the lake contains no water, sets the area to 
the maximum possible when the lake has reached its maximum volume (O_max_a) at 
overspill, and otherwise calculates the area based on the volume. 
 
Each lake area converter needs a similar expression. 
 



3) Next, what do you need to do to allow the lakes to overflow when they fill up 
with water?  Write the equation for overflow that you’re using for Owens 
Lake and apply the same logic to the other lakes in the chain. 

 
You need to put in an if-then statement that compares the volume in each lake to the 
maximum volume the lake is capable of holding, and that will overflow the excess to the 
next lake in the chain if the volume exceeds the maximum volume.  For example: 
 
overflow_to_china= 
if(Owens_volume>25.e9)then(Owens_volume-25.e9)else(0.0) 
 
If Owens Lake's volume exceeds 25.*109 m3, the excess (Owens_volume-25.e9) is allowed 
to flow to the China lake.  Otherwise, there is no outflow. 
 
4) Use the data in the “hypsometry” kaleidagraph file to determine the 

relationship between depth and volume for each lake. 
 
Just as with the lake area converter, the lake depth converter requires a statement to limit 
the depth of the lake to its maximum depth at spillover.  For example, for Owens lake, the 
following statement is used: 
 
Owens_depth= 
if(Owens_volume<=0.)then(0.)else 
if(Owens_volume>=25.e9)then(O_max_d)else(3.5837+(4.009e-9*Owens_volume)-
(9.5021e-20*(Owens_volume^2))+(1.2319e-30*(Owens_volume^3)))   
 
This statement sets the depth to zero when the lake contains no water, constrains the 
depth to the depth at spillover when the lake is at its maximum volume, and calculates 
intermediate depths from the polynomial function of the volume. 
 
5) Fill in your remaining initial conditions – initial lake volumes = 0, and 

evaporation and runoff rates as specified in your reading for the modern 
climate. 

 
6) Run your model for about 200 years and describe and explain the resulting 

behavior of the lake level curves (depth).  What size lakes do you get?  Why 
does the depth of Owens Lake eventually reach steady state?  

Under modern climatic conditions, the only lake to contain water is Owens Lake.  It 
reaches a steady state depth of ~16-17 m. 
 
Owens Lake reaches steady state once the water coming in is balanced by the water going 
out via evaporation.  This balance is achieved as soon as the lake has grown to a 
sufficiently large surface area that the area multiplied by the evaporation rate equals the 
amount of runoff entering the lake. 
 
7) How long does it take for Owens Lake to reach 95% of its steady state depth?  

Note:  I’m asking for 95% of steady state because the lake approaches steady 
state asymptotically, so it is difficult to determine when it has reached 
complete steady state.  For all future questions regarding time required to 
reach steady state you may also look at the 95% value. 

 
About 53 years. 
 
8) Experiment with changing the amount of runoff into the lake.  Double, triple, 

and halve the runoff.  What impact does runoff amount have on the time 
required for Owens lake to reach steady state? 

 
With a halving of runoff, it takes ~12 years to reach a steady state depth of 5.8 m. 

With a doubling of runoff, it takes the lake ~150 years to reach steady state depth of 58 m. 

With a tripling of runoff, it takes ~44 years to reach steady state, but this number is 
misleading.  In fact, the lake is now overflowing to China lake, which in turn is 
overflowing into Searles lake.  For this reason, it would be more appropriate to look at the 
response of all 3 lakes. 

In general, however, as long as the lake is not overflowing, an increase in runoff leads to 
an increase in the amount of time required to reach steady state.  
 
9) Change your runoff back to the modern value and then experiment with 

changing evaporation rates.  What impact does evaporation rate have on the 
time required to reach steady state? 

 
Dropping Owens Lake's evaporation rate to 1.0 m/yr from 1.34 m/yr results in a time to 
reach steady state of ~115 years and a depth of 29 m. 

Increasing the evaporation rate to 1.6 m/yr, drops the time to reach steady state to ~25 
years and the depth to ~12 m.   

In general, an increase in evaporation rate decreases the time required to get to steady 
state.  Another way to think about this is that increasing the evaporation rate allows the 
evaporative outflow to balance the inflow more quickly. 
 
10) Change your evaporation rate back to the modern value.  Let’s mess around 

with changing the area/volume relationship for Owens Lake to see what 
impact lake hypsometry (basin shape) has on the response time for the lake 
(response time = time required to reach steady state). 

 
Change your exponent in the area/volume equation to values higher and lower 
than the value you got for Owens Lake.  What happens to the response time as 
you change the exponent?  Why? 
 
The response time is extremely sensitive to changes in the area/volume relationship.  
Dropping the exponent on the area/volume equation to 0.35 from 0.37636 results in a 52 
m deep lake (as opposed to 16-17 m) that reaches 95% of its steady state volume after 
~280 years.  

Raising the value of the exponent to 0.4, results in a lake only 7.3 m deep that reaches 
95% of its steady state volume in 11.5 years. 

The explanation for this behavior lies in an understanding of what the exponent means in 
the overall lake geometry.  A lake with a high value for the exponent has a flat, pan shape.  
As volume increases, surface area increases markedly.  On the other hand, a lake with a 



low value for the exponent has a narrow, cone shape.  As volume increases, surface area 
changes little.   
 
Because of the high surface area to volume ratio, evaporation from the flat pan-shaped 
lake quickly balances inflow, allowing the lake to reach a steady state rapidly.  In 
contrast, evaporation from the narrow, cone-shaped lake doesn't change much as runoff 
fills the lake, and a long time is required before the evaporative outflow balances the 
inflow. 
 
Students who use exponents lower than 0.35 will find that they get a turn around in the 
response time.  In other words, the lower the exponent, the faster the lake appears to reach 
steady state.  You should point out to them, however, that this apparent steady state 
results solely from the fact that the lake is overflowing, and that to really gauge how long 
it takes for the lake system to come to steady state, they must include the lakes 
downstream. 
 
11) Set your exponent back to its original value.  Now determine how much you 

need to increase runoff by in order to get water to flow all the way to Lake 
Manly given modern evaporation rates.   

 
A minimum of 9* modern runoff is required to get water to Manly given the model I've 
included in this website.  At this level of runoff, the water that flows into Manly is 
completely evaporated in every timestep, leading to oscillatory behavior.  A value of 10* 
modern runoff leads to an ~23 m deep lake at the end of the run. 

12) We know that glacial period evaporation rates were significantly depressed 

relative to modern because of the colder average surface temperature of 
Earth.  Incorporate a 30% reduction in evaporation rate to all of the lakes and 
again determine how much water is required to get water to spill into Lake 
Manly. 

 
To carry out this experiment, multiply the evaporation outflow for each lake by 0.7.  For 
example, for Owens Lake: 
 
evap_from_Owens= 
1.34*0.7*Owens_area 
 
where 1.34 is the modern evaporation rate. 
 
Doing this for all of the lakes in the chain reduces the amount of runoff necessary to get 
overflow to Manly to about 7 times modern: 

13) How long does it take the lake chain to reach a steady state under this glacial 

scenario?   
 
Manly stabilizes at a depth of ~23 m in ~300 years.  At this point, the whole chain is in 
steady state. 
 



14) Given the response times you have determined in 12 and 13, comment on the 
ability of the lake chain to record climatic changes (such as long droughts or 
exceptionally wet periods) that occur on the timescale of decades. 

 
Since it takes the lake chain on the order of hundreds of years to respond to a change in 
climate, it is unlikely that it would be able to record changes that occur on the timescale 
of decades.  The lake chain would still be responding to the last change when it was forced 
to respond to the new change.  These short climatic events would therefore either not be 
recorded or only partially recorded. 
 
15) Would the lakes be able to record climatic changes that occur on the timescale 

of thousands of years? 
 
Since the response time is on the order of hundreds of years, the lake chain should be able 
to faithfully record climatic changes occurring on longer timescales. 
 
16) Using your modern evaporation rates, let’s put your answers to 14 and 15 to 

the test.  Create an equation that will allow the runoff to vary between 
modern and 10x modern values with a period of 100 years.  What is your 
equation? 

 
runoff_from_sierras =  
5.5*3.98e8+4.5*(3.98e8*sin(2*pi*time/100)) 
 
When the sin has a value of -1, the runoff will be 1*modern where modern = 3.98e8.  
When the sin has a value of +1, the runoff will be 10*modern. 
 
17) Run the model for several hundred years and show the behavior of the lakes.  

Is the chain in steady state?  How do you know? 
 
With a period of 100 years, the lake chain is never in steady state.  This can be determined 
by examining the behavior of Owens Lake carefully.  The climate passes through modern 
conditions every 100 years, yet Owens Lake never achieves its modern depth of 16.5 m at 
these times.  The lake level drops, but doesn't fall below about 55 m.  The climate is 
changing so rapidly that the lake chain doesn't have a chance to fall to its modern levels 
before it has to respond to a wetter climate: 

18) Now change the period to 1000 years and run the model for a few thousand 

years.  Is the chain in steady state now? 
  
The chain is much closer to being in steady state.  This time, Owens Lake drops to ~20 m 
during episodes of modern climate (as compared to ~16.5 m in the steady state case for 
modern climate), suggesting the chain is responding at close to steady state. 

The following plot was created using a period of 2500 years and a timestep of 1/4 year.  
In this scenario, Owens Lake comes down to its modern steady state value of ~16.5 m 
during modern climate. 

